




හයවන ලිපිය - තෙවන කොටස (III)

0 සිට 99 දක්වා ගිණිය හැකි ගණක පරිපථයක් නිර්මාණය කරමු

පසුගිය ලිපියෙන් අප ඉදිරිපත් කළේ 0 සිට 9 දක්වා ගිණිය හැකි ගණක පරිපථයක් නිර්මාණය කරගන්නා ආකාරයයි. එය මදක් වැඩිදියුණු කොට 0 සිට 99 දක්වා හෙවත් අංක දෙකකින් යුත් ගණක පරිපථයක් නිර්මාණය කරන අයුරු මෙම ලිපියෙන් ඉදිරිපත් කෙරේ. රූප සටහන අංක 1 මගින් අදාළ පරිපථය දැක්වේ.



රූප සටහන අංක 1

Digit 1 හා Digit 2 ලෙස දක්වා ඇති සජන බණ්ඩා ප්‍රදර්ශක දෙකේ a සිට g දක්වා වූ අනු මයික්‍රොකොන්ට්‍රොලරයේ RB0 සිට RB6 දක්වා වූ අනුවලට සම්බන්ධ කර ඇති අතර GND හෙවත් පොදු අනු D 400 ව්‍යාප්තීන්ට දෙකක් හරහා සැපයුණේ (-) අනුයට සවි කර ඇත. මයික්‍රොකොන්ට්‍රොලරයේ RA1 හා RA2 අනු මගින් එම ව්‍යාප්තීන්ට පාලනය වේ. RA1 අනුයේ වෝල්ටීයතාව 5V බවට පත් වූ විට එනම් PORTA රෙජිස්ටරයේ 1 වැනි බිටුව තාර්කික 1 වූ විට ව්‍යාප්තීන්ට තුළින් ධාරාව ගැලීම සිදු වේ. එනම් PORTB රෙජිස්ටරයට ලියන අගයට අදාළ ඉලක්කම Digit 1 මගින් දර්ශනය වේ. එලෙස RA2 අනුයේ වෝල්ටීයතාව 5V වූ විට Q2 තුළින් ධාරාව ගැලීම නිසා Digit 2 මගින් PORTB හි ඇති අගයට අදාළ ඉලක්කම ප්‍රදර්ශනය කෙරේ. RA1 හා RA2 යන අනු දෙකෙන් සැම විට ම එකක් පමණක් 5V තත්ත්වයේ තිබිය යුතු අතර අනෙක 0V තත්ත්වයේ විය යුතු ය. එසේ නො වී RA1 හා RA2 යන අනු දෙක ම 0V වුව හොත් සජන බණ්ඩා ප්‍රදර්ශක දෙක ම නිවී යන අතර කිසි ම ඉලක්කමක් දිස් නො වේ. එමෙන්ම RA1 හා RA2 අනු දෙක ම 5V වුව හොත් Digit 1 හා Digit 2 යන ප්‍රදර්ශක දෙක ම එක අංකයක් ප්‍රදර්ශනය කරයි.

එ අනුව නිසි ක්‍රියාකාරීත්වය ලබා ගැනීමට නම් වරකට එක ප්‍රදර්ශකය බැගින් Digit 1 හා Digit 2 ප්‍රදර්ශක මාරුවෙන් මාරුවට ක්‍රියාත්මක කළ යුතු ය. මෙය තවදුරටත් උද්ගරණයකින් තොරව ගැනීමට උත්සාහ කරමු. ඔබට 25 නමැති අංකය ප්‍රදර්ශනය කළ යුතු නම්, Digit 1 මගින් "5" ඉලක්කම ද Digit 2 මගින් "2" ඉලක්කම ද ප්‍රදර්ශනය කළ යුතු වේ. මේ සඳහා මුලින් ම "5" ඉලක්කමට අදාළ

බිටු සැකැස්ම වන "01101101" PORTB රෙජිස්ටරයට ලිවිය යුතු ය. ඉන් පසු Digit 1 හි පොදු අගය සම්බන්ධ කර ඇති Q₁ ව්‍යාප්තීන්ට තුළින් ධාරාව ගැලීමට සැලැස්විය යුතු ය. එ සඳහා RA1 අනුය 5V තත්ත්වයට ගෙන ආ යුතු ය. PORTA රෙජිස්ටරයේ "1" වැනි බිටුවට "1" ලිවීමෙන් එය සිදු කරගත හැකි ය. එවිට "5" ඉලක්කම Digit 1 මගින් ප්‍රදර්ශනය වේ. ඉන් පසුව ඇත්තේ අනෙක් ඉලක්කම වන දෙවන ඉලක්කම Digit 2 මගින් ප්‍රදර්ශනය කිරීම ය. එසේ කිරීමට නම් කළ යුත්තේ එම ඉලක්කමට අදාළ බිටු සැකැස්ම PORTB රෙජිස්ටරයට ලියා Q₂ ව්‍යාප්තීන්ට තුළින් ධාරාව ගැලීමට සැලැස්වීම ය. එහෙත් මෙවිට Q₁ ව්‍යාප්තීන්ට තුළින් ද ධාරාව ගැලීමට හැකි ව තිබුණේ නම් Digit 1 හි ද පසේ ඉලක්කම වෙනුවට දෙවන ඉලක්කම දිස් වේ. එම නිසා වරකට ප්‍රදර්ශනය කළ යුත්තේ එක් ඉලක්කමක් පමණි. එ අනුව Digit 1 මගින් පසේ ඉලක්කම ප්‍රදර්ශනය කරන විට Digit 2 මගින් කිසි ම ඉලක්කමක් ප්‍රදර්ශනය නො කළ යුතු අතර Digit 2 මගින් දෙවන ඉලක්කම ප්‍රදර්ශනය කරන විට Digit 1 මගින් කිසිදු ඉලක්කමක් ප්‍රදර්ශනය නොවිය යුතු ය.

එ අනුව Digit 1 හා Digit 2 මගින් අදාළ ඉලක්කම් දෙක වෙන වෙන ම අවස්ථා දෙකක දී ප්‍රදර්ශනය කරයි. මෙම ක්‍රියාවලිය නොතවත්වා වේගයෙන් කරගත යාමේ දී ප්‍රදර්ශකය නිවී ඇති අවස්ථාව අපට ග්‍රහණය නො වී ඉලක්කම් දෙක ම දැල්වී ඇති ආකාරයක් පෙනේ. එ අනුව 0 සිට 99 දක්වා වූ ඕනෑ ම අංකයක් ප්‍රදර්ශනය කර ගත හැකි ය. මෙම සංකල්පය ඉලක්කම් හතරක් හෝ පහක් සඳහා වුව ද වලංගු වේ. පරිපථය හා ක්‍රමලේඛනය සංකීර්ණ වීම වළක්වා ගැනීම සඳහා ඉලක්කම් දෙකකට පමණක් සීමා කිරීමට අප සිදු වුවත් මදක් උත්සාහ කළ හොත් ඉලක්කම් දෙකකට වඩා

වැඩි ප්‍රමාණයක් ප්‍රදර්ශනය කිරීම සඳහා පරිපථය හා ක්‍රමලේඛනය වෙනස් කර ගැනීම ඔබට ම කරගත හැකි වනු ඇතැයි අපි විශ්වාස කරමු.

මෙම පරිපථයේ දී ද විදුලිය සැපයූ විගස ම Digit 1 හා Digit 2 මගින් බිංදුව ප්‍රදර්ශනය කරයි. බොත්තම වරක් එබූ විට Digit 2 මගින් බිංදුව ද Digit 1 මගින් එක ද ප්‍රදර්ශනය කෙරේ. එ අනුව බොත්තම ඔබන වාරයක් පාසා ප්‍රදර්ශනය කෙරෙන අගය එකකින් වැඩි වේ. අවසානයේ දී "99"ට පැමිණ නැවත බිංදුවෙන් පටන් ගනී. බොත්තම වෙනුවට වෙනත් යුදස්ස සංවේදකයක් භාවිත කර මෙම පරිපථය ම දෙරටුවක් හරහා ගමන් කරන මිනිසුන් ගණන ගණන් කිරීම සඳහා යොදාගත හැකි ය. මෙම ලිපි පෙළේ අරමුණ වනුයේ මයික්‍රොකොන්ට්‍රොලර පිළිබඳව ඔබ දන යුතු මූලික කරුණු ගෙනහැර දැක්වීම වන අතර එම දැනුම සහ ඔබ ගේ නිර්මාණශීලී හැකියාවන් භාවිත කර නව නිර්මාණ බිහි කිරීම ඔබට හැර වේ.

දැන් අදාළ ක්‍රමලේඛන දෙකට හැරෙමු. රූප සටහන අංක 2 න් දැක්වෙනුයේ එම ක්‍රමලේඛනය වන අතර සුපුරුදු පරිදි මුලින් ම රෙජිස්ටර හා විචල්‍යයන් හැඳින්වීම සිදු කර ආරම්භයේ දී ම එ බව ලබා දිය යුතු අගයන් ද ලබා දී ඇත.

ඉන්පසු ** display the first digit ** යන සටහනට පසුව තිබෙන Loop යන ලේඛයට ඉදිරියෙන් ඇති bcf PORTA, 2 යන උපදෙසෙන් කියනුයේ PORTA රෙජිස්ටරයේ දෙවැනි බිටුව බිංදුව කළ යුතු ය යන්නයි. එවිට RA2 අගයේ වෝල්ටීයතාව 0V බවට පත් වී Q₂ ව්‍යාප්තීන්ට තුළින් ධාරාව නො ගලන තත්ත්වයක් ඇති වේ. ඉන් පසු movf Digit 1, 0 යන උපදෙසින් කියවෙනුයේ Digit 1 හි දිස්

```

;****Define Registers****
PC          equ      02h
STATUS      equ      03h
PORTA       equ      05h
PORTB       equ      06h
TRISA       equ      85h
TRISB       equ      86h

```

```

CounterL    equ      0Dh
CounterH    equ      0Eh
Digit1      equ      10h
Digit2      equ      11h

```

```

;****initialization****
bsf          STATUS,5      ;Switch to Bank 1
movlw        b'00111001'
movwf        TRISA         ;RA0 I/P RA1,2 O/P
clrf         TRISB         ;PORT B output
bcf          STATUS,5      ;Switch to Bank 0
clrf         Digit1        ;initilize to zero
clrf         Digit2        ;initilize to zero

```

```

; **display the first digit**
Loop         bcf          PORTA,2      ;disable digit2
              movf        Digit1,0     ;get digit1 value
              call        Table        ;Look Up Table
              movwf       PORTB        ;Writ the number
              bsf         PORTA,1      ;enable digit1

```

```

; **small delay to display digit1**
Delay1       decfsz      CounterL,1
              goto        Delay1

```

```

; **display the second digit**
bcf          PORTA,1      ;disable digit1
movf         Digit2,0     ;get digit2 value
call         Table        ;Look Up Table
movwf        PORTB        ;Writ the number
bsf          PORTA,2      ;enable digit2

```

```

; **small delay to display digit2**
Delay2       decfsz      CounterL,1
              goto        Delay2

btfss        PORTA,0      ;Button pressed?
goto         Countup      ;If yes Count up
goto         Loop         ;else back to Loop

```

```

; **Update Digit1 and Digit2**
Countup      incf         Digit1,1     ;increase Digit1
              movlw       D'10'       ;load w with ten
              xorwf        Digit1,0   ;compare with 10
              btfss        STATUS,2   ;if equal result 0
              goto         Loop       ;else back to Loop
              clrf         Digit1     ;make Digit1 = 0
              incf         Digit2,1   ;increase Digit1
              movlw       D'10'       ;load w with ten
              xorwf        Digit2,0   ;compare with 10
              btfsc        STATUS,2   ;if equal result 0
              clrf         Digit1     ;make Digit2 = 0
              goto         Loop       ;back to Loop

```

```

;****Look Up Table for bit patterns****
Table        addwf        PC,1
              retlw        b'00111111' ;Number 0
              retlw        b'00000110' ;Number 1
              retlw        b'01011011' ;Number 2
              retlw        b'01001111' ;Number 3
              retlw        b'01100110' ;Number 4
              retlw        b'01101101' ;Number 5
              retlw        b'01111101' ;Number 6
              retlw        b'00000111' ;Number 7
              retlw        b'01111111' ;Number 8
              retlw        b'01100111' ;Number 9

end

```

රූප සටහන අංක 2

විය යුතු අගය W රෙජිස්ටරයට කොපි කළ යුතු ය යන්න ය. ඉන්පසු Call table යනු එම ඉලක්කමට අදාළ බිටු සැකැස්ම ලබා දෙන වගුව වෙත ගොස් එය W රෙජිස්ටරයට ලබාගෙන නැවත movwf

වැනි පිටුවට

මොරටුව විශ්ව විද්‍යාලයේ විද්‍යුත් හා විදුලි සංදේශ අංශයේ ගාමිණී ජයසිංහ/කෝලින ධර්මප්‍රිය